Q:

Open Specific RDP connection using "IF" or "GOTO" in Batch file

I have tried multiple things and everything fails. I am constantly having to log in to various terminal service with RDP and want to be able to control them all with a .bat. The following is my latest failed attempt. When I run it, it opens both connections.
@Echo off

Rem Sets the parameter for the network id.
set /p Server = "Server> "
Set User = "User" 
Set Pass = "Password" 

If "%Server%" =="233" (
  GoTo 233
) Else If "%Server%" =="234" (
  GoTo 234
)

Rem RDP Connection to Server 234
:234

cmdkey /generic:TERMSRV/RDC234 /user:%User% /pass:%Pass%
  mstsc /v:RDC234

Rem RDP Connection to Server 233
:233

cmdkey /generic:TERMSRV/RDC233 /user:%User% /pass:%Pass%
  mstsc /v:RDC233

Pause

A:

@Echo off

Rem Sets the parameter for the network id.
set /p Server="Server> "
Set "User=User" 
Set "Pass=Password" 

If "%Server%"=="233" goto valid
If "%Server%"=="234" goto valid
echo Invalid server "%server%" specified
goto end

:valid    
Rem RDP Connection to valid Server
cmdkey /generic:TERMSRV/RDC%server% /user:%User% /pass:%Pass%
  mstsc /v:RDC%server%

:END

Pause

It's not clear from your code whether you're actually inputting the values for user and pass (which should follow the server pattern) or you are setting user and pass to values you've suppressed.
Either way, spaces on each side of the = are significant, so your code sets "userSpace" not "user" for instance.
Using the quotes in the manner I have shown will ensure that the value assigned does not include any trailing spaces present on the line. As you posted it originally, the value assigned would include the quotes.
It's quite illogical to use an if condition goto else... If the condition is false, the next line will be executed, so
if condition goto ...
if anothercondition goto ...
rem get here if neither first nor second condition.

would seem to be far clearer.
Since the insructions to be executed in :233 and :244 are identical, bar the server value, you can use %server% in the manner shown to simply butt-up the value of the fixed-text prefix rdc.

